
Intel Fortran Compiler 11.0 



Agenda 

Fortran 2003 features 

Other language features 

New and changed behaviors 



11.0 Features from Fortran 2003 

•  Enumerators 

•  Type extension (not polymorphic) 

• Allocatable scalars (not deferred-length character) 

•  ERRMSG keyword for ALLOCATE and DEALLOCATE 

• SOURCE= keyword for ALLOCATE 

• MAX/MIN/MAXVAL/MINVAL/MAXLOC/MINLOC with 
CHARACTER arguments 

•  Intrinsic modules IEEE_EXCEPTIONS, 
IEEE_ARITHMETIC and IEEE_FEATURES 

• ASSOCIATE construct 



11.0 F2003 Features contd. 

•  PROCEDURE declaration 

•  Procedure pointers 

• ABSTRACT INTERFACE 

•  PASS and NOPASS attributes 

• Structure constructors with component names and 
default initialization 

• Array constructors with type and character length 
specifications 

• BLANK, DELIM, ENCODING, IOMSG, PAD, ROUND, 
SIGN I/O keywords 

• DC, DP, RD, RC, RN, RP, RU, RZ format edit 
descriptors 



Enumerators 

Another way to define named integer constants 

ENUM, BIND(C) 

  ENUMERATOR GREEN 

  ENUMERATOR :: RED = 4, BLUE = 9 

  ENUMERATOR YELLOW 

END ENUM 

GREEN is 0, YELLOW is 10 

Exactly as if PARAMETER was used 

Integer kind is same as C_INT (4) 

Interoperable with C enumerators 

:: required if value specified 



Type Extension 

Define a type that adds components to a “base type” 

The base type must be “extensible” 

•  Not a SEQUENCE type 

•  Not a BIND type 

Extended type inherits all components of its base type 

Base type can be host- or use-associated 

Extended type can itself be extended 

Base type becomes itself a component of extended type 



Type Extension 

TYPE POINT 

  REAL :: X, Y 

END TYPE POINT 

TYPE, EXTENDS(POINT) :: COLOR_POINT 

  ! X and Y and component name POINT inherited 

  INTEGER :: COLOR 

END TYPE COLOR_POINT 

TYPE(COLOR_POINT) :: CP 

CP%X, CP%Y, CP%COLOR defined 

CP%POINT, CP%POINT%X, CP%POINT%Y also defined! 



Allocatable Scalars 

Scalar (non-array) variables can be ALLOCATABLE 

integer, allocatable :: i 

type mytype 

  real :: a = 3.14 

  real :: b = 2.17 

end type mytype 

type(mytype), allocatable :: r 

allocate (i) 

allocate (r) 

Deferred-length character not yet supported 

CHARACTER(LEN=:) FOO; ALLOCATE(CHARACTER(20)::FOO) 



ERRMSG for ALLOCATE/DEALLOCATE 

New ERRMSG keyword for ALLOCATE and DEALLOCATE 

CHARACTER variable assigned implementation-defined message if 
error 

allocate (x(5), stat=stat, errmsg=message) 

print *, stat, trim(message) 

… 

 151 allocatable array is already allocated 



ALLOCATE (SOURCE=) 

ALLOCATE (Y(10), SOURCE=X) 

In F2003, X must be “conformable” with Y (same shape or a scalar) 

After allocation, Y is assigned value of X 

Gets dynamic type of X, but dynamic types not yet implemented 

F2008 (draft) removes “conformable” restriction and says Y is 
allocated with shape of X (bounds for X omitted and ranks must 
match) 

We implement F2008 behavior 



MAX/MIN/MAXVAL/MINVAL/MAXLOC/MINLOC 
with CHARACTER arguments 

In F95, these were limited to numeric types 

character(1) :: arr(7) = ['f','o','r','t','r','a','n'] 

print *, min('bcd','abdf','hhhh') 

print *, minloc(arr) 

print *, maxval(arr) 

… 

 abdf 

           6 

 t 



IEEE Intrinsic Modules 

Provides information about and some control of the IEEE floating point 
environment 

Defines derived types, constants, procedures and operators 

Intrinsic modules IEEE_EXCEPTIONS, IEEE_ARITHMETIC, 
IEEE_FEATURES 

IEEE_ARITHMETIC uses IEEE_EXCEPTIONS 

If IEEE_EXCEPTIONS or IEEE_ARITHMETIC visible then 
IEEE_OVERFLOW and IEEE_DIVIDE_BY_ZERO are supported 



IEEE_EXCEPTIONS 

IEEE_FLAG_TYPE derived type for exceptions 

IEEE_STATUS_TYPE derived type for status values 

Exception values IEEE_INVALID, IEEE_OVERFLOW, 
IEEE_DIVIDE_BY_ZERO, IEEE_INEXACT 

IEEE_USUAL array of OVERFLOW, DIVIDE_BY_ZERO, INVALID 

IEEE_ALL array of all exceptions 

Procedures to get and set flags, status, halting mode, support 



IEEE_ARITHMETIC 

IEEE_CLASS_TYPE to identify what class a value is (denorm, infinity, 
NaN, etc.) 

IEEE_ROUND_TYPE to identify a rounding mode 

Support inquiry functions for IEEE datatypes, denorms, divide, infinity, 
NaN, rounding, formatting, square root, underflow 

Elemental functions class, copy sign, is finite, is NaN, is normal, round 
to integer and more 

Get and set routines for flags and modes 

== and /= operators for the derived types (in case you want to 
compare) 



IEEE_FEATURES 

Defines “features” that you want supported 

•  IEEE_DATATYPE, IEEE_DENORMAL, IEEE_DIVIDE, IEEE_HALTING, 
IEEE_INEXACT, IEEE_INEXACT_FLAG, IEEE_INF, 
IEEE_INVALID_FLAG, IEEE_NAN, IEEE_ROUNDING, IEEE_SQRT, 
IEEE_UNDERFLOW_FLAG 

USE, INTRINSIC :: IEEE_FEATURES, &  
  ONLY: IEEE_INVALID_FLAG, IEEE_HALTING 

Behavior is implementation-dependent if IEEE_FEATURES not visible 

Portable equivalent of switches such as -fpe 



USE, INTRINSIC :: IEEE_EXCEPTIONS 
USE, INTRINSIC :: IEEE_FEATURES, ONLY: IEEE_INVALID_FLAG 
! The other exceptions of IEEE_USUAL (IEEE_OVERFLOW and 
! IEEE_DIVIDE_BY_ZERO) are always available with IEEE_EXCEPTIONS 
TYPE(IEEE_STATUS_TYPE) STATUS_VALUE 
LOGICAL, DIMENSION(3) :: FLAG_VALUE 
... 
CALL IEEE_GET_STATUS(STATUS_VALUE) 
CALL IEEE_SET_HALTING_MODE(IEEE_USUAL,.FALSE.) ! Needed in case the 
! default on the processor is to halt on exceptions 
CALL IEEE_SET_FLAG(IEEE_USUAL,.FALSE.) 
! First try the "fast" algorithm for inverting a matrix: 
MATRIX1 = FAST_INV(MATRIX) ! This shall not alter MATRIX. 
CALL IEEE_GET_FLAG(IEEE_USUAL,FLAG_VALUE) 
IF (ANY(FLAG_VALUE)) THEN 
! "Fast" algorithm failed; try "slow" one: 
  CALL IEEE_SET_FLAG(IEEE_USUAL,.FALSE.) 
  MATRIX1 = SLOW_INV(MATRIX) 
  CALL IEEE_GET_FLAG(IEEE_USUAL,FLAG_VALUE) 
  IF (ANY(FLAG_VALUE)) THEN 
    WRITE (*, *) ’Cannot invert matrix’ 
    STOP 
    END IF 
  END IF 
CALL IEEE_SET_STATUS(STATUS_VALUE) 

Example of using IEEE modules 



ASSOCIATE Construct 

Associates a name with a variable or expression in a section of a 
program 

ASSOCIATE (Z => EXP(-X**2+Y**2))*COS(THETA)) 

  PRINT *, A+Z, A-Z 

  END ASSOCIATE 

Not a macro nor like a statement function 

ASSOCIATE  (XC => A%B(I,J)%C) 

  XC%DV = XC%DV + PRODUCT(XC%EV(1:N)) 

  END ASSOCIATE 

ASSOCIATE (ARRAY => AX%B(I,:)%C) 

  ARRAY(N)%EV = ARRAY(N-1)%EV 

  END ASSOCIATE 

Array bounds are LBOUND:LBOUND+SIZE-1 



PROCEDURE 

procedure () :: foo 

•  foo is either a function or subroutine 

•  parentheses required 
procedure(real) :: foo2 

•  foo2 is a function that returns a REAL value 
procedure(sqrt) :: foo3 

•  foo3 has the same interface as intrinsic SQRT 
procedure(somefunc) :: foo4 

•  foo4 has the same interface as somefunc, which 
can be EXTERNAL, or have an explicit interface 

PROCEDUREs always have the EXTERNAL attribute 



PROCEDURE (2) 

subroutine somesub (procarg) 

procedure(integer), optional :: procarg 

•  procarg is an optional procedure argument 
procedure(someint), bind(C,NAME=“JoJo”) :: & 
billy 

•  billy is a procedure with the same interface as 
“someint” and is interoperable with C with an 
external name of “JoJo” 



Procedure Pointers 

procedure(someint), pointer :: pp => null() 

pp => AnotherProc 

call pp(3.14) 

Procedure pointers can have SAVE attribute if locals, INTENT if dummy 
argument 

Target of procedure pointer assignment must have matching interface 



ABSTRACT INTERFACE 

A way to declare an interface without it corresponding to an external 
procedure 

abstract interface 
integer function fuzzy (x) 
real x 
end function fuzzy 

end interface 

procedure(fuzzy), pointer :: dice 



Procedure Pointer Components 

A procedure pointer can be a component of a derived type 

type mytype 

procedure(someint), pointer :: p => null() 

real :: r 

end type mytype 

When procedure pointer component p is called, an extra argument is 
passed with the derived type value for context (default is as first 
argument) 



PASS, NOPASS 

Control of passing context in call to procedure pointer 

Use only for component of derived type 

TYPE MYTYPE 
INTEGER CONTEXT 
PROCEDURE(INT1),PASS,POINTER :: A 
PROCEDURE(INT2),NOPASS,POINTER :: B 
PROCEDURE(INT3),PASS(ARG),POINTER :: C 

END TYPE MYTYPE 

•  Call through A passes extra first argument of type MYTYPE 
(default) 

•  Call through B does not pass extra argument 

•  Call through C passes type(MYTYPE) through argument ARG 



Putting it together (1) 
module mymod 

type mytype 

integer i 

procedure (foobar), pointer, pass :: bar 

end type mytype 

abstract interface 

  subroutine foobar (a,b) 

  import 

  type(mytype) :: a 

  integer b 

  end subroutine foobar 

end interface 



Putting it together (2) 
contains 

subroutine modfoobar (a,b) 

type(mytype) :: a 

integer b 

print *, a%i, b 

return 

end subroutine modfoobar 

end module mymod 

subroutine myfoobar (a,b) 

use mymod 

type(mytype) :: a 

integer b 

print *, a%i, 2*b 

return 

end subroutine myfoobar 



Putting it together (3) 

program pass 

use mymod 

type (mytype) :: r 

procedure(foobar) :: myfoobar 

r%i = 314 

r%bar => modfoobar 

call r%bar(5) 

r%bar => myfoobar 

call r%bar(6) 

end program pass 



Putting it together (4) 

Running the program gives: 

        314           5 

        314          12 



C Interop and Procedure Pointers 

Derived type C_FUNPTR interoperable with C function pointer type 

Function C_FUNLOC(X) returns type C_FUNPTR with pointer to 
procedure X 

•  X must be an interoperable procedure or procedure pointer 

Subroutine C_F_PROCPTR(CPTR,FPTR) associates Fortran procedure 
pointer FPTR with the target of C function pointer CPTR (type 
C_FUNPTR) 



Structure Constructor Enhancements 

Old way – structure constructor was just a list of values and had to 
name all elements 

•  mytype(1,2,3) 

New way – can use component name keywords and can omit values 
that have default initialization 

type mytype 
integer a,b 
real :: c = 0.0 
end type mytype 

… 

r = mytype(4,b=5) ! c is defaulted 



Array constructor enhancements 

Can now use type specification, including character length 

•  (/INTEGER(8) :: 1,2,3/) 
•  [CHARACTER(LEN=7) :: 'Red','Green','Blue','Orange'] 
In F95, value character lengths must all be the same (though we 
supported differing lengths) 

If intrinsic type specified, all values must be conformable to the type 

If derived type specified, all values must be of that type 

Note square bracket syntax, which is newly standard in F2003, though 
we’ve supported it “forever” 



New I/O Keywords 

BLANK, DELIM, PAD, ROUND, SIGN temporarily override OPEN 
specifiers in I/O control lists 

ROUND=UP,DOWN,ZERO,NEAREST,COMPATIBLE, 
PROCESSOR_DEFINED in OPEN,INQUIRE,I/O control list 

ENCODING=UTF-8, DEFAULT on OPEN (only DEFAULT supported) 

SIGN=PLUS,SUPPRESS,PROCESSOR_DEFINED 

IOMSG for all I/O statements 

SIZE= on INQUIRE returns size of file in “storage units” (always bytes) 

•  Unit must be opened, otherwise -1 returned 



New FORMAT edit descriptors 

DC Decimal Comma 

DP Decimal Period 

RD Round Down 

RC Round Compatible 

RN Round Nearest 

RP Round Processor-Defined 

RU Round Up 

RZ Round towards Zero 



Still to come 

Full object-oriented features including CLASS, dynamic types, finalizers 

Type-bound procedures 

PUBLIC types with PRIVATE components and PRIVATE types with 
PUBLIC components 

Parameterized derived types 

User-defined derived type I/O 

Support for international character sets 



Non F2003 Features in 11.0 

OpenMP 3.0 

•  !$OMP TASK – create an asynchronous task 

•  !$OMP TASKWAIT – wait for an asynchronous task 

•  Nested parallelism 

•  SCHEDULE AUTO and OMP_SCHEDULE environment variable 

UNROLL_AND_JAM and NOUNROLL_AND_JAM directives 

•  Similar to UNROLL 

VECTOR NONTEMPORAL can name variables 

VECTOR TEMPORAL (prevent optimizer from setting VECTOR 
NONTEMPORAL) 



Other Changes 

ifortvars.bat now takes arguments (ia32, intel64, ia32_intel64, 
ia32_ia64, ia64) 

Visual Studio projects must be converted to 11.0 format 

SSE2 now default on IA-32 

Lots of new optimization switches 

IMSL refreshed, but not new version 

VS08 Shell replacing VSPPE deferred to EMHE 

idb gone on Windows 

fpp is documented  



More to Explore 

Fortran 2003 Standard: 
http://ssgsites.sc.intel.com/sites/SSG-SPD-ICL/support/Shared
%20Documents/Fortran%202003%20Standard.pdf 

The New Features of Fortran 2003 

http://www.fortranplus.co.uk/resources/john_reid_new_2003.pdf  





New Features for Intel Fortran in 11.1 

Arriving June 2009 



Summary of new features 

• Fortran 2003 features 

• Miscellaneous improvements 

• Correction of a >30-year old mistake 

• Visual Studio 2008 Shell replaces VSPPE 

F2003: CLASS declaration 

• CLASS is a variant of TYPE that can match a type and any extension 
of that type 

• Valid for pointer, allocatable and dummy argument only 

TYPE :: point 
  REAL :: x 
  REAL :: y 
END TYPE point 
TYPE, EXTENDS(point) :: point_3D ! In 11.0 
  REAL :: z 
END TYPE point_3D 

TYPE(point) :: p1 ! Matches only point 
CLASS(point) :: p2 ! Matches point and point_3D 

CLASS (contd.) 
• Type of object declared with CLASS is “declared type” 

• Type of actual object associated is “dynamic type” 

• Object declared with CLASS is “polymorphic”; can have more than one 
type 

•  Restriction: You can access components of the declared type only, 
not any extended type (but see SELECT TYPE later) 

• CLASS(*) is “unlimited polymorphic” – matches any derived type 

•  Cannot access any components without SELECT TYPE 

•  Can ALLOCATE with desired type: 
•  ALLOCATE(TYPE(FOO) :: P) 

• EXTENDS_TYPE_OF(A,B) – logical, is dynamic type of A an extension 
of the dynamic type of B? 

• SAME_TYPE_AS(A,B) – logical, is dynamic type of A the same as that 
of B? 

F2003: SELECT TYPE 

• Like SELECT CASE for dynamic type 

• Within the CLASS_IS or TYPE_IS block can access components of that 
type/class 

SELECT TYPE ( A => P_OR_C ) 
CLASS IS ( POINT ) 
! "CLASS ( POINT ) :: A" implied here 
PRINT *, A%X, A%Y  
TYPE IS ( POINT_3D ) 
! "TYPE ( POINT_3D ) :: A" implied here 
PRINT *, A%X, A%Y, A%Z 
CLASS DEFAULT 
! Something else 
END SELECT 

F2003: Type-bound procedures 

• Create methods for objects of a type or class 

TYPE POINT 
REAL :: X, Y 
CONTAINS 
PROCEDURE, PASS :: LENGTH => POINT_LENGTH 
END TYPE POINT 
TYPE(POINT) :: P,Q 

CONTAINS 

REAL FUNCTION POINT_LENGTH (THIS, B) 
CLASS (POINT), INTENT (IN) :: THIS, B 
POINT_LENGTH = SQRT ((THIS%X - B%X)**2 + (THIS%Y - B%Y)**2 ) 
END FUNCTION POINT_LENGTH 

• Call with P%LENGTH(Q) 

Type-Bound Procedures (contd.) 

• Type-bound procedure can also be GENERIC – not supported in 11.1 

• No concept of “constructor” – do it yourself 

• FINAL subroutines (“finalizer”) not supported in 11.1 

F2003: Overriding 

TYPE POINT 
REAL :: X, Y 
CONTAINS 
PROCEDURE, PASS :: LENGTH => POINT_LENGTH 
END TYPE POINT 

TYPE, EXTENDS(POINT) :: POINT_3D 
REAL :: Z 
CONTAINS 
PROCEDURE, PASS :: LENGTH => POINT3D_LENGTH ! Overrides 
END TYPE POINT_3D 

• NON_OVERRIDABLE keyword disables overriding 

• Rules similar to generics for compatibility 

F2003: Abstract Type 

An abstract type is a type which must be extended to be usable 

TYPE, ABSTRACT :: DRAWABLE_OBJECT 
REAL, DIMENSION(3) :: RGB_COLOR=(/1.0,1.0,1.0/) ! White 
REAL, DIMENSION(2) :: POSITION=(/0.0,0.0/) ! Centroid 
CONTAINS 
PROCEDURE(RENDER_X), PASS(OBJECT), DEFERRED :: RENDER 
END TYPE DRAWABLE_OBJECT 

ABSTRACT INTERFACE ! In 11.0 
SUBROUTINE RENDER_X(OBJECT, WINDOW) 
CLASS(DRAWABLE_OBJECT), INTENT(IN) :: OBJECT 
CLASS(X_WINDOW), INTENT(INOUT) :: WINDOW 
END SUBROUTINE RENDER_X 
END INTERFACE 

Abstract Type (contd.) 

TYPE, EXTENDS(DRAWABLE_OBJECT) :: & 
   DRAWABLE_TRIANGLE ! Not ABSTRACT 
REAL, DIMENSION(2,3) :: VERTICES  
CONTAINS 
PROCEDURE, PASS(OBJECT) :: RENDER=>RENDER_TRIANGLE_X 
END TYPE DRAWABLE_TRIANGLE 

• Abstract types and interfaces are patterns 

F2003: Deferred-length Character 

• Special case of allocatable scalar, not in 11.0 

• Finally – variable-length character strings come to Fortran! 

CHARACTER(:), ALLOCATABLE :: C 
ALLOCATE (CHARACTER(20) :: C) 
PRINT *, LEN(C) ! Prints 20 

F2003: Selective accessibility of 
components 

• Ability to declare PUBLIC types with PRIVATE components or vice-
versa 

TYPE, PUBLIC :: FOO 
INTEGER :: A ! Public 
PRIVATE ! Everything after this default PRIVATE 
REAL :: B ! Private 
COMPLEX, PUBLIC :: C ! Public 
END TYPE FOO 

F2003: Miscellaneous 

• NAMELIST I/O allowed on internal file (character variable) 

• Automatic array allowed in a NAMELIST group 

• MAXLOC/MINLOC return zero for zero-size array 

•  Requires option /assume:noold_maxminloc to be set 

•  Performance will suffer – default is that result is implementation-
dependent (F95 definition) 

• COUNT_RATE argument to SYSTEM_CLOCK may be REAL 

F2003 with 11.1: What’s Missing? 

• FINAL and GENERIC type-bound procedures 

• User-defined derived type I/O 

• Parameterized derived types 

Fortran 2003 References: 

•  http://www.fortranplus.co.uk/resources/john_reid_new_2003.pdf 

•  http://j3-fortran.org/doc/2003_Committee_Draft/04-007.pdf  

5/11/09 Intel® Fortran 11.1 New Features 51 



Non-F2003 changes 

• When bounds checking is on, length of passed character argument 
used for checking if less than declared length 

• LOGICAL input values (T, F, .TRUE., .FALSE.) no longer accepted for 
list-directed and NAMELIST input of non-LOGICAL values 

•  Corrects mistake made by DEC over 30 years ago 

•  Old behavior enabled with /assume:old_logical_ldio 

• /fpe-all for per-compilation control of floating point exception 
environment (instead of main program only) 

• Windows: New static library ifmodintr.lib contains constants and 
routines from intrinsic modules 

•  Fixes problem where C_NULL_PTR, etc., have wrong values when 
DLL linked. Not needed on Linux/Mac 

Visual Studio 2008 Shell 

• New VS environment for standalone Windows Fortran 

• Replaces Visual Studio 2005 Premier Partner Edition (VSPPE) 

• VSPPE not recognized by 11.1 integration – will install VS08 shell if no 
other supported VS 

•  If VSPPE and VS03 or VS08 are installed, 11.1 upgrade will remove 
integration from VSPPE – can be replaced with custom install 

• No significant difference in appearance or usage – same limitations 
apply: 

•  No resource editor 

•  No conversion of CVF projects 

